

# SCIENCE & VIE

MENSUEL

N° 848 MAI 1988

• Cancer :  
l'escroquerie  
à l'interleukine

• La navette spatiale  
soviétique

• Un moteur d'auto  
révolutionnaire



**TAPIS VERT:  
5000 MILLIONNAIRES  
DE TROP!**

BELGIQUE 120 FB - SUISSE 5 FS - CANADA \$ 3,50

ESPAGNE 500 Ptas - MAROC 18 Dh - TUNISIE 2,3 Dt - ITALIE 4000 L - USA NYC \$ 3,75 - OTHER \$ 3,95

M 2578 - 848 - 17,00 F



3792578017005 08480

# Des illustrations musicales

## INFORMATIQUE AMUSANTE

**N**ous vous proposons un récapitulatif des illustrations musicales utilisées lors de nos programmes précédents. Nous les avons réparties suivant diverses sous-routines. Il sera donc possible de les réutiliser directement pour réaliser l'accompagnement musical d'un programme que vous aurez créé. Chaque sous-routine comportera donc l'ensemble des éléments nécessaires à l'exécution de la musique choisie.

Voyons l'écriture de notre programme. Il comportera une première partie faisant office de page d'accueil et également chargée de l'envoi aux sous-routines. Ensuite, nous trouverons 6 sous-routines. 5 d'entre elles concerneront l'exécution des musiques et la dernière sera utilisée pour obtenir, éventuellement, le listing de chacune d'elles.

Notre page de présentation est affichée par les lignes 100 à 280 du programme. Les lignes 500 à 540 sont, elles, réservées à l'aiguillage du programme vers chaque sous-routine. Nous formerons donc une boucle FOR... NEXT autour de la variable Q et nous lui associerons une instruction ON GOSUB. Notons également que l'ordre INKEY\$ de la ligne 505 permet d'interrompre l'exécution du programme dès qu'une touche est frappée au clavier, de manière à pouvoir obtenir le listing de l'illustration musicale choisie.

La première de nos sous-routines concerne la musique du programme "LEONCE" et sera placée des lignes 1000 à 1180 du programme. Les enveloppes sonores seront mises en place, puis les données concernant la mélodie mémorisées par les chaînes DATA des lignes 1030 et 1040. Le reste de cette sous-routine sera consacré à l'exécution de la musique. Nous trouverons donc trois instructions SOUND placées dans une boucle



FOR... NEXT. La lecture des notes sera effectuée par la ligne 1090. Notons que la ligne 1095 provoque l'interruption de cette sous-routine lorsqu'une touche est tapée au clavier.

Les quatre sous-routines suivantes sont basées sur le même principe. Nous retrouverons donc toujours des chaînes DATA pour la mémorisation de la mélodie, des ordres SOUND pour son exécution et une ligne permettant d'interrompre cette dernière sur la frappe d'une touche au clavier.

La dernière sous-routine, débutant à la ligne 8000 aura pour mission d'afficher le listing de la musique choisie. Précisons qu'après son utilisation, l'Amstrad interrompra définitivement l'exécution du programme. Pour le relancer, il faudra taper RUN puis ENTER, tout comme lors de sa première exécution.

La frappe ne doit pas poser de problème particulier. De même, l'utilisation des diverses sous-routines proposées dans d'autres programmes, est relativement sim-

**ILLUSTRATIONS MUSICALES.**

- 1: LEONCE.
- 2: SYNTHETISEUR.
- 3: LAPIN BLEU.
- 4: PALET.
- 5: JAZZ.

MUSIQUE EN COURS: LEONCE.  
NUMERO CHOISI? 2

POUR OBTENIR LE LISTING  
D'UNE MELODIE TAPER UNE TOUCHE.

```

100 CLS: CLEAR
103 REM *****
104 REM *
105 REM * PAGE DE PRESENTATION *
106 REM *
107 REM *****
200 LOCATE 5,3: PRINT "ILLUSTRATIONS MUSICALES."
210 LOCATE 5,8: PRINT "1: LEONCE."
220 LOCATE 5,10: PRINT "2: SYNTHETISEUR."
230 LOCATE 5,12: PRINT "3: LAPIN BLEU."
240 LOCATE 25,8: PRINT "4: PALET."
250 LOCATE 25,10: PRINT "5: JAZZ."
260 LOCATE 5,15: PRINT "MUSIQUE EN COURS:"
270 LOCATE 5,20: PRINT "POUR OBTENIR LE LISTING "
280 LOCATE 5,22: PRINT "D'UNE MELODIE TAPER UNE TOUCHE."
500 FOR Q=1 TO 5
501 REM *****
502 REM * INTERRUPTION SUR TOUCHE FRAPPEE *
503 REM *****
505 IF INKEY$("<>") THEN GOSUB 8000
506 REM *****
507 REM * AIGUILLAGE VERS LES SOUS-ROUTINES *
508 REM *****
510 ON Q GOSUB 1000,2000,3000,4000,5000
520 FOR H=1 TO 1000:NEXT H
530 NEXT Q
540 GOTO 500
901 REM *****
902 REM *
903 REM * MUSIQUE DU PROGRAMME "LEONCE" *
904 REM *
905 REM *****
1000 LOCATE 25,15:PRINT "LEONCE." :RELEASE 7
1007 REM *****
1008 REM * DETERMINATION DES ENVELOPPES SONORES *
1009 REM *****
1010 ENV 1,1,12,1,1,0,10,6,-2,1
1020 ENV 2,3,5,1,1,0,15,5,-3,1
1027 REM *****
1028 REM * DONNEES DE LA MELODIE *
1029 REM *****
1030 DATA 402,13,602,3,804,5,902,7,1012,9,902,11,1012,9,802,11
1040 DATA 402,5,602,3,804,5,902,7,1012,9,902,11,1012,9,1204,11
1047 REM *****
1048 REM * EXECUTION *
1049 REM *****
1050 FOR P=0 TO 2
1060 FOR T=2 TO 1 STEP -1
1070 RESTORE 1030
1080 FOR I=0 TO 15
1090 READ X:READ B
1091 REM *****
1092 REM * INTERRUPTION SUR TOUCHE FRAPPEE *
1093 REM *****
1095 IF INKEY$("<>") THEN RETURN
1100 FOR S=0 TO 100 :NEXT S
1110 SOUND 1,202,0,0,1,0,0
1120 SOUND 2,301,0,0,1,0,0
1130 IF SQ (2)<>132 THEN GOTO 1130
1140 SOUND 3,INT(X/T),0,0,2,0,0
1150 NEXT I

```

```

1160 NEXT T
1170 NEXT P
1180 RETURN
1901 REM *****
1902 REM *
1903 REM * MUSIQUE DU PROGRAMME "BRUITS" *
1904 REM *
1905 REM *****
2000 LOCATE 25,15:PRINT "SYNTHETISEUR."
2007 REM *****
2008 REM * DONNEES DE LA MELODIE *
2009 REM *****
2010 DATA 239,190,478,379,239,190,358,268
2020 FOR K=1 TO 2
2027 REM *****
2028 REM * DETERMINATION DES ENVELOPPES SONORES *
2029 REM *****
2030 ENV 6,1,15,2,1,-3,6,12,-1,1
2037 REM *****
2038 REM * EXECUTION DE LA MUSIQUE *
2039 REM *****
2040 LET P=5:LET W=1
2050 FOR I=1 TO 2
2060 FOR O=1 TO 2
2070 FOR L=1 TO 2
2080 RESTORE 2010
2090 FOR M=0 TO 3
2091 REM *****
2092 REM * INTERRUPTION SUR TOUCHE FRAPPEE *
2093 REM *****
2095 IF INKEY$("<>") THEN RETURN
2100 READ N1:READ N2
2110 SOUND 3,12,2,P,0,0,1
2120 SOUND 1,N1/W,0,0,6,0,0
2130 SOUND 2,N2/(O*L*W),0,0,6,0,0
2140 NEXT M
2150 NEXT L
2160 NEXT O
2170 NEXT I
2180 NEXT K
2190 RETURN
2901 REM *****
2902 REM *
2903 REM * MUSIQUE DU PROGRAMME "LAPIN BLEU" *
2904 REM *
2905 REM *****
3000 LOCATE 25,15:PRINT "LAPIN BLEU."
3001 REM *****
3002 REM * DETERMINATION DES ENVELOPPES SONORES *
3003 REM *****
3005 ENV 1,1,15,1,1,0,1,15,-1,1
3010 ENV 4,1,15,1,5,-1,3,10,-1,2
3020 ENV 5,5,3,1,1,0,1,1,-15,1
3027 REM *****
3028 REM * DONNEES DE LA MELODIE *
3029 REM *****
3030 DATA 1,239,1,253,4,239,4,478,4,478,1,319,1,358,1,379,1,319,1,239,1,253,4,239
3040 DATA 1,213,1,225,1,213,4,426,4,426,1,284,1,253,1,284,4,319
3050 DATA 1,239,1,253,4,239,4,478,4,478,1,319,1,358,1,379,1,319,1,239,1,253,4,239
3060 DATA 1,213,1,225,1,213,4,426,1,239,1,253,1,284,1,253,4,239

```



ple. Il faudra éventuellement les renuméroter, puis les insérer à l'aide de l'ordre MERGE et veiller à ce qu'aucune des variables utilisées porte le même nom que celles du programme principal.

L'utilisation de ce programme reste également très simple. Après l'avoir entièrement tapé et demandé RUN, l'ensemble des titres des illustrations musicales s'affichera sur l'écran. Ces dernières seront alors exécutées successivement. Notons que l'écran précisera en

permanence le nom de la musique en cours d'exécution. La frappe d'une touche provoque l'interruption du programme. L'ordinateur demandera alors le numéro correspondant à la mélodie dont vous souhaitez connaître le listing. Ce chiffre sera frappé au clavier, puis validé par ENTER. Dès lors, le listing sera affiché.

Notons enfin que, pour éviter d'avoir à recopier la sous-routine souhaitée à chaque fois pour l'introduire dans un nouveau pro-

gramme, il sera parfaitement possible d'utiliser la fonction DELETE de l'Amstrad puis MERGE.

Si, par exemple, on souhaite utiliser uniquement la musique exécutée en 1000 mais avec pour numéro de sous-routine 7000, il sera possible de procéder comme suit. Après chargement du programme on tapera DELETE 1000, puis DELETE 1200, puis RENUM 7000, 1000, 10 et enfin MERGE "XXXX" ; XXXX étant le nom du nouveau programme. **Henri-Pierre Penel**

```

3067 REM *****
3068 REM * EXECUTION DE LA MUSIQUE *
3069 REM *****
3070 FOR G=2 TO 4 STEP 2
3080 RESTORE 3030
3090 FOR I=0 TO 41
3091 REM *****
3092 REM * INTERRUPTION SUR TOUCHE FRAPPEE *
3093 REM *****
3095 IF INKEY$("<>") THEN RETURN
3100 READ E:READ N
3110 SOUND 1,N+3,0,0,E,0,0
3120 SOUND 2,N/G,0,0,E,0,0
3130 NEXT I
3140 NEXT G
3150 RETURN
3901 REM *****
3902 REM *
3903 REM * MUSIQUE DU PROGRAMME "PALET" *
3904 REM *
3905 REM *****
4000 LOCATE 25,15:PRINT "PALET."
4001 REM *****
4002 REM * DETERMINATION DES ENVELOPPES SONORES *
4003 REM *****
4005 ENV 1,1,15,1,5,-2,1,5,-1,1
4010 ENV 2,1,15,2,5,-3,2,5,0,2
4020 ENV 3,1,15,1,5,-1,1,5,-1,3
4027 REM *****
4028 REM * EXECUTION DE LA MUSIQUE *
4029 REM *****
4030 FOR Z=1 TO 3
4040 FOR I=1 TO 4
4050 RESTORE 4060
4057 REM *****
4058 REM * DONNEES DE LA MELODIE *
4059 REM *****
4060 DATA 0,1,2,0,1,2,0,2,15,0,1,2,956,1,0,0,2,15
4070 FOR M=1 TO 6
4075 IF INKEY$("<>") THEN GOTO 8000
4080 READ N:READ F:READ BR
4090 SOUND 1,N,0,0,F,0,0,BR
4100 NEXT M
4110 NEXT I
4120 FOR I=1 TO 3
4130 RESTORE 4140
4140 DATA 1,0,1,2,2,956,3,0,1,0,1,2
4150 DATA 2,676,3,0,1,0,2,15,2,638,3,0
4160 DATA 1,0,1,2,0,0,0,0,1,1276,1,0
4170 DATA 2,758,3,0,1,0,2,15
4180 FOR M=1 TO 11
4181 REM *****
4182 REM * INTERRUPTION SUR TOUCHE FRAPPEE *
4183 REM *****
4185 IF INKEY$("<>") THEN RETURN
4190 READ C:READ N:READ F:READ BR
4200 SOUND C,N,0,0,F,0,0,BR
4210 NEXT M
4220 NEXT I
4230 FOR I=1 TO 3
4240 RESTORE 4250
4250 DATA 1,0,1,2,2,30,1,0,1,0,1,2
4260 DATA 2,60,1,0,1,0,2,15,2,60,2,0
4270 DATA 1,0,1,2,2,30,1,0,1,956,1,0
4280 DATA 2,30,1,0,1,0,2,15,2,60,2,0
4290 FOR M=1 TO 12
4291 REM *****
4292 REM * INTERRUPTION SUR TOUCHE FRAPPEE *
4293 REM *****
4295 IF INKEY$("<>") THEN RETURN
4300 READ C:READ N:READ F:READ BR
4310 SOUND C,N,0,0,F,0,0,BR
4320 NEXT M
4330 NEXT I
4340 NEXT Z
4350 RETURN
4901 REM *****
4902 REM *
4903 REM * MUSIQUE DU PROGRAMME "JAZZ" *
4904 REM *
4905 REM *****
5000 LOCATE 25,15:PRINT "JAZZ."
5005 FOR H=1 TO 2
5007 REM *****
5008 REM * DONNEES DE LA MELODIE *
5009 REM *****
5010 DATA 284,379,451,568,451,379,284,568
5011 REM *****
5012 REM * DETERMINATION DES ENVELOPPES SONORES *
5013 REM *****
5020 ENV 1,1,15,5,1,-3,1,12,-1,2
5030 ENV 2,1,12,1,1,0,7,12,-1,1
5040 ENV 3,1,10,1,1,0,4,5,-2,1
5050 ENV 4,1,12,1,1,0,5,12,-1,2
5057 REM *****
5058 REM * EXECUTION DE LA MUSIQUE *
5059 REM *****
5060 FOR B=1 TO 4
5080 FOR T=1 TO 2
5090 RESTORE 5010
5100 FOR I=1 TO 4
5101 REM *****
5102 REM * INTERRUPTION SUR TOUCHE FRAPPEE *
5103 REM *****
5105 IF INKEY$("<>") THEN RETURN
5110 READ X
5120 IF SQ(2)<>4 THEN GOTO 5120
5130 IF B>1 THEN SOUND 1,X,0,0,1,0,0
5140 SOUND 2,0,0,0,2,0,1
5150 READ X
5160 SOUND 2,0,0,0,3,0,1
5170 SOUND 2,0,0,0,4,0,1
5180 IF B>1 THEN SOUND 1,X*T,0,0,1,0,0
5190 NEXT I
5200 NEXT T
5210 NEXT B
5220 NEXT H
5230 RETURN
7001 REM *****
7002 REM *
7003 REM * SOUS-ROUTINE DE LISTING *
7004 REM *
7005 REM *****
8000 LOCATE 5,17:INPUT "NUMERO CHOISI";K
8010 IF K=1 THEN LIST 1000-1900
8020 IF K=2 THEN LIST 2000-2900
8030 IF K=3 THEN LIST 3000-3900
8040 IF K=4 THEN LIST 4000-4900
8050 IF K=5 THEN LIST 5000-5900
8060 RETURN

```